

Практическое занятие №

Тема: «Подключение и программирование модуля джойстика»

Цель работы: приобрести практические навыки по подключению и программированию модуля джойстика к Arduino.

Последовательность выполнения работы:

- Собрать схемы на макетной плате, иначе при отсутствии набора Arduino в web-приложениях (<https://wokwi.com/projects/new/arduino-uno> или <https://www.tinkercad.com/>) для приведенных примеров.
- Запрограммировать микроконтроллер согласно заданию в примере.

Содержание отчета:

- название практического занятия, его цель;
- фото или скриншоты собранной схемы;
- написанный программный код вставить текстом, Courier New, 12 кегль, одинарный отступ без абзацев;
- вывод о проделанной работе;
- файл Fritzing с принципиальной и монтажной схемой.

ТЕОРЕТИЧЕСКИЕ СВЕДЕНИЯ

Аналоговый джойстик состоит из двух потенциометров (ось X и Y) и кнопки (ось Z). При перемещении ручки значения сопротивления меняются, что позволяет считывать аналоговые напряжения.

Ось X: горизонтальное перемещение (0-1023 на Arduino).

Ось Y: вертикальное перемещение (0-1023 на Arduino).

Кнопка: нажатие джойстика (логический 0 или 1).

Принцип работы:

В нейтральном положении значения X и Y близки к 512.

При отклонении ручки значения меняются:

- **X_{min}** (лево) → 0, **X_{max}** (право) → 1023.
- **Y_{min}** (низ) → 0, **Y_{max}** (верх) → 1023.

Кнопка замыкает цепь при нажатии.



Рисунок 1 – Модуль аналогового джойстика

ЗАДАНИЕ

Схема:




Рисунок 2 – Схема кодключения

Программа:

```
// Пины для светодиодов
const int ledUp = 3;    // Вверх
const int ledDown = 4; // Вниз
const int ledLeft = 5; // Лево
const int ledRight = 6; // Право
const int ledButton = 7; // Светодиод кнопки

// Пин для кнопки джойстика
const int buttonPin = 2;

// Аналоговые пины для осей X и Y
const int xAxisPin = A0;
const int yAxisPin = A1;

// Порог чувствительности
const int threshold = 100;

void setup() {
    // Настройка пинов светодиодов как выходов
    pinMode(ledUp, OUTPUT);
    pinMode(ledDown, OUTPUT);
    pinMode(ledLeft, OUTPUT);
    pinMode(ledRight, OUTPUT);
    pinMode(ledButton, OUTPUT);

    // Настройка пина кнопки с подтяжкой к питанию
    pinMode(buttonPin, INPUT_PULLUP);

    // Инициализация последовательного порта
    Serial.begin(9600);
    Serial.println("Джойстик инициализирован!");
}

void loop() {
    // Чтение значений осей X и Y
    int xValue = analogRead(xAxisPin);
    int yValue = analogRead(yAxisPin);

    // Чтение состояния кнопки
    bool buttonState = !digitalRead(buttonPin);

    // Управление светодиодом кнопки
```

```
//использование тернарного оператора: УСЛОВИЕ ? ИСТИНА : ЛОЖЬ
digitalWrite(ledButton, buttonState ? HIGH : LOW);

// Сбрасываем все светодиоды
analogWrite(ledUp, 0);
analogWrite(ledDown, 0);
analogWrite(ledLeft, 0);
analogWrite(ledRight, 0);

// Обработка направлений и установка яркости
if (yValue < (512 - threshold)) {
    // Вверх
    int brightness = map(yValue, 512 - threshold, 0, 0, 255);
    analogWrite(ledUp, brightness);
}
else if (yValue > (512 + threshold)) {
    // Вниз
    int brightness = map(yValue, 512 + threshold, 1023, 0, 255);
    analogWrite(ledDown, brightness);
}

if (xValue < (512 - threshold)) {
    // Лево
    int brightness = map(xValue, 512 - threshold, 0, 0, 255);
    analogWrite(ledLeft, brightness);
}
else if (xValue > (512 + threshold)) {
    // Право
    int brightness = map(xValue, 512 + threshold, 1023, 0, 255);
    analogWrite(ledRight, brightness);
}

// Вывод в монитор порта
Serial.print("X: ");
Serial.print(xValue);
Serial.print(" Y: ");
Serial.print(yValue);
Serial.print(" Кнопка: ");
Serial.print(buttonState ? "НАЖАТА" : "отпущена");
```

```
Serial.print(" | ");

// Определение направления для вывода
if (xValue < (512 - threshold) && yValue < (512 -
threshold)) {
    Serial.println("Вверх-Лево");
} else if (xValue > (512 + threshold) && yValue < (512 -
threshold)) {
    Serial.println("Вверх-Право");
} else if (xValue < (512 - threshold) && yValue > (512 +
threshold)) {
    Serial.println("Вниз-Лево");
} else if (xValue > (512 + threshold) && yValue > (512 +
threshold)) {
    Serial.println("Вниз-Право");
} else if (yValue < (512 - threshold)) {
    Serial.println("Вверх");
} else if (yValue > (512 + threshold)) {
    Serial.println("Вниз");
} else if (xValue < (512 - threshold)) {
    Serial.println("Лево");
} else if (xValue > (512 + threshold)) {
    Serial.println("Право");
} else {
    Serial.println("Нейтрально");
}

// Задержка
delay(100);
}
```